МИНИСТЕРСТВО ОБРАЗОВАНИЯ РЕСПУБЛИКИ БЕЛАРУСЬ БЕЛОРУССКИЙ ГОСУДАРСТВЕННЫЙ УНИВЕРСИТЕТ
Факультет прикладной математики и информатики

Сергиенко Лев Эдуардович
Отчет по Лабораторная работа 3

Преподаватель
Волчецкая П.С.
Дрепакова А.В.

Минск, 2025

Формулировка проблемы

В некоторой стране имеются N нефтяных вышек, i-я из которых способна выкачать до a_i тысяч баррелей нефти в сутки, и M нефтеперерабатывающих заводов, каждый из которых, в свою очередь, способен переработать до b_j тысяч баррелей нефти в сутки. Заводы и вышки связаны трубопроводами; ввиду различных причин стоимость транспортировки одной тысячи баррелей от i-й вышки к j-му заводу составляет c_{ij} денежных единиц; объем перекачки не ограничен. Учитывая, что вышки и заводы должны находиться в производственном балансе, т. е. вся выкачиваемая нефть должна быть вовремя переработана (излишки нигде не должны сливаться или накапливаться), какое максимальное количество нефти в сутки может перерабатывать данная инфраструктура? На данный вопрос ответить относительно легко, поэтому нас интересует лишь величина минимальных возможных затрат среди всех максимальных по объему перерабатываемой нефти решений.

Математическая модель

Модель — линейная, стандартная транспортная задача. Можно решать:

- линейным программированием (симплекс / транспортный метод),
- как min-cost max-flow в сетевой модели (добавляем источник s, сток t, ребра s→вышки с пропускной способностью a_i, ребра заводы→t с b_j, ребра вышки→завод с бесконечной пропускной способностью и стоимостью c_{ij}; затем ищем максимальный поток минимальной стоимости).

Параметры

- (N, M) — числа вышек и заводов;
- (a_i) для $(i = 1 \dots N)$ — предложение (тыс. барр.);
- (b_j) для $(j = 1 \dots M)$ — спрос (тыс. барр.);
- (c_{ij}) — стоимость перевозки 1 тыс. барр. с (i)-й вышки на (j)-й завод.

Переменные

- $(x_{ij} \geq 0)$ — объем (в тыс. барр.) поставки с вышки (i) на завод (j).

Целевая функция

$$\text{minimize} ; Z = \sum_{i=1}^{N} \sum_{j=1}^{M} c_{ij}, x_{ij}$$

Ограничения

$$\forall i : \quad \sum_{j=1}^{M} x_{ij} = a_i$$

вся нефть, добытая на вышке (i), должна быть отправлена

$$\forall j : \quad \sum_{i=1}^{N} x_{ij} = b_j$$

завод (j) принимает ровно (b_j)

$$x_{ij} \geq 0$$

Поскольку $(\sum_i a_i = \sum_j b_j)$, система равенств совместна и задает полный (максимальный) поток между источниками и стоками. Минимизация стоимости при этих ограничениях дает искомое решение.

Решение

Задача решена с применением языка математического моделирования AMPL с использованием оптимизационного решателя CPLEX.

Модель (oil.mod)

```
param N >= 1 integer; # Количество вышек
param M >= 1 integer; # Количество заводов

param a{1..N} >= 0; # Производительность вышек
param b{1..M} >= 0; # Производительность заводов
param c{1..N, 1..M} >= 0; # Стоимость транспортировки

var flow_source{1..N} >= 0; # Поток от источника к вышкам
var flow_sink{1..M} >= 0; # Поток от заводов к стоку
var flow{1..N, 1..M} >= 0; # Поток от вышек к заводам
```

```
# Целевая функция: минимизация стоимости транспортировки
minimize total_cost:
sum{i in 1..N, j in 1..M} c[i,j] * flow[i,j];

# Ограничение на производительность вышек
subject to well_capacity{i in 1..N}:
flow_source[i] <= a[i];

# Ограничение на производительность заводов
subject to factory_capacity{j in 1..M}:
flow_sink[j] <= b[j];

# Баланс потока в вышках
subject to well_balance{i in 1..N}:
flow_source[i] = sum{j in 1..M} flow[i,j];

# Баланс потока в заводах
subject to factory_balance{j in 1..M}:
sum{i in 1..N} flow[i,j] = flow_sink[j];

# Общий поток должен быть максимальным = min(sum(a), sum(b))
subject to max_flow:
sum{i in 1..N} flow_source[i] = min(sum{i in 1..N} a[i], sum{j in 1..M}
b[j]);
```

Данные (oil.dat)

```
data;

param N := 3;
param M := 4;

param a :=
    1  3
    2  6
    3  7;

param b :=
    1  2
    2  5
    3  1
    4  8;

param c: 1  2  3  4 :=
```

```
1    1  2  3  4
2    8  7  6  5
3    9 12 10 11;
```

Запуск (oil.run)

```
model oil.mod;
data oil.dat;

option solver cplex;

solve;

printf "Minimum transportation cost: %d\n", total_cost;
printf "Total processing volume: %.0f thousand barrels\n\n", sum{i in
1..N} flow_source[i];

printf "Flows from wells to factories:\n";
printf "%-10s", "Well\\Factory";
for {j in 1..M} {
    printf "%8s%d", "Fact", j;
}
printf "\n";

for {i in 1..N} {
    printf "Well %d     ", i;
    for {j in 1..M} {
        printf "%8.2f", flow[i,j];
    }
    printf "\n";
}

printf "\nExtraction from wells:\n";
for {i in 1..N} {
    printf "Well %d: %.2f / %d\n", i, flow_source[i], a[i];
}

printf "\nDelivery to factories:\n";
for {j in 1..M} {
    printf "Factory %d: %.2f / %d\n", j, flow_sink[j], b[j];
}
```

Результат

Решение задачи из условия:

```
ampl: include "oil.run";
CPLEX 22.1.1.0: optimal solution; objective 110
7 dual simplex iterations (0 in phase I)
Minimum transportation cost: 110
Total processing volume: 16 thousand barrels

Flows from wells to factories:
Well\\Factory    Fact1    Fact2    Fact3    Fact4
Well 1          0.00    3.00    0.00    0.00
Well 2          0.00    0.00    0.00    6.00
Well 3          2.00    2.00    1.00    2.00

Extraction from wells:
Well 1: 3.00 / 3
Well 2: 6.00 / 6
Well 3: 7.00 / 7

Delivery to factories:
Factory 1: 2.00 / 2
Factory 2: 5.00 / 5
Factory 3: 1.00 / 1
Factory 4: 8.00 / 8
```

Минимальные затраты на транспортировку: **110** (ден. ед.).

Общий объем переработки: **16** тыс. баррелей (равен сумме aia_iai и bjb_jbj).

Матрица потоков (тыс. барр.):

	Fact1	Fact2	Fact3	Fact4
Well 1	0.00	3.00	0.00	0.00
Well 2	0.00	0.00	0.00	6.00
Well 3	2.00	2.00	1.00	2.00

Это соответствует выводу приведенного в условии примера.