КУРС "ИССЛЕДОВАНИЕ ОПЕРАЦИЙ"

Лабораторная работа No 2

Тема:
Линейное программирование.
Решение задач оптимизации производства

Вариант 33

Сергиенко Лев 12 группа МСС, 4 курс

Формулировка проблемы

В обувном цехе для изготовления трех моделей обуви используются четыре вида комплектующих материалов, запасы которых, расход их на изготовление одной пары обуви и цены, получаемые от реализации пары обуви, приведены в таблице.

Составьте математическую модель задачи и найдите оптимальное решение.

Запасы сырья и нормы их расхода на производство обуви

Комплектующие	Расход комплектующих на изготовление обуви			Запасы
	сапоги	ботинки	ботильоны	
Кожа	0,2	0,3	0.32	2 700
Каблуки	0.5	0,2	0.45	800
Стельки	0.027	0.022	0.21	1 600
Подошвы	0.1	0.25	0.28	100
Цена ед.. руб.	900	500	700	

Это классическая задача линейного программирования - задача планирования производства при ограниченных ресурсах.

Переменные - объёмы выпуска (действительные, непрерывные), ограничения - линейные по ресурсам, цель - линейная (максимизация прибыли/выручки).

Математическая модель ЛП

Переменные:

- x_1 - число пар сапог;
- x_2 - число пар ботинок;
- x_3 - число пар ботильонов.

Целевая функция (максимизация выручки):

$Z = 900x_1 + 500x_2 + 700x_3 -> max$

Ограничения (ресурсы):

$$
\begin{aligned}
0.2x_1 + 0.3x_2 + 0.32x_3 &\le 2700 \\
0.5x_1 + 0.2x_2 + 0.45x_3 &\le 800 \\
0.027x_1 + 0.022x_2 + 0.21x_3 &\le 1600 \\
0.1x_1 + 0.25x_2 + 0.28x_3 &\le 100 \\
x_1,\ x_2,\ x_3 &\ge 0
\end{aligned}
$$

Добавляя $s_1..s_4 \ge 0$, получаем систему равенств:

$$
\begin{aligned}
0.2x_1 + 0.3x_2 + 0.32x_3 + s_1 &= 2700 \\
0.5x_1 + 0.2x_2 + 0.45x_3 + s_2 &= 800 \\
0.027x_1 + 0.022x_2 + 0.21x_3 + s_3 &= 1600 \\
0.1x_1 + 0.25x_2 + 0.28x_3 + s_4 &= 100
\end{aligned}
$$

Это стандартная LP-задача (m=4, n=3).

Файлы .mod, dat, .run,

.mod

```
param m integer >= 1;
param n integer >= 1;

param A{1..m, 1..n}; # матрица потребления ресурсов
param b{1..m}; # запасы ресурсов
param price{1..n}; # цена (выручка) с единицы

# переменные: целые числа пар каждой модели
var x{1..n} integer >= 0;

# целевая: максимум выручки
maximize Revenue: sum{j in 1..n} price[j] * x[j];

# ресурсные ограничения
s.t. Resource{i in 1..m}:
sum{j in 1..n} A[i,j] * x[j] <= b[i];
```

.dat

```
data;
param m := 4;
param n := 3;

param A :
      1      2      3  :=
1   0.2    0.3   0.32   # кожа
2   0.5    0.2   0.45   # каблуки
3  0.027  0.022   0.21  # стельки
4   0.1    0.25  0.28   # подошвы
;

param b :=
1 2700   # кожа
2  800   # каблуки
3 1600   # стельки
4  100   # подошвы
;

param price :=
1 900
2 500
3 700
;
end;
```

.run

```
option solver cplex;
model shoes.mod;
data shoes.dat;
solve;
printf "\n===== Result =====\n";
display Revenue, x;
```

Результат

Запуск алгоритма дал:

- Оптимальное решение (вектор всех переменных: $x_1, x_2, x_3, s_1, s_2, s_3, s_4$):

x = [1000.   0.   0. 2500. 300. 1573.   0.]

То есть:

- $x_1 = 1000$ (сапоги)
- $x_2 = 0$ (ботинки)
- $x_3 = 0$ (ботильоны)

Оптимальная выручка: Z = 900 000 руб.

Проверка ограничений при $x_1 = 1000, x_2 = x_3 = 0$:

- кожа: $0.2 \cdot 1000 = 200 \leq 2700$
- каблуки: $0.5 \cdot 1000 = 500 \leq 800$
- стельки: $0.027 \cdot 1000 = 27 \leq 1600$
- подошвы: $0.1 \cdot 1000 = 100 = 100$ - подошвы полностью использованы.

```
CPLEX 22.1.1.0: optimal integer solution; objective 900000
0 MIP simplex iterations
0 branch-and-bound nodes

===== Result =====
Revenue = 9e+05

x [*] :=
1  1000
2     0
3     0
;
```